$\alpha+\omega$, 2002, Nr. 2, 78–82

Trigonometrinių funkcijų reikšmių simbolinis skaičiavimas

Aleksas Domarkas
aleksas@ieva.mif.vu.lt

Mokykloje sužinome, kam lygūs kampų $\frac{\pi}{6}$, $\frac{\pi}{4}$, $\frac{\pi}{3}$, $\frac{\pi}{2}$ sinusai. Jų reikšmės yra arba racionalieji skaičiai, arba iracionalieji skaičiai, užrašomi pasinaudojus kvadratine šaknimi. Straipsnyje pateikiamos $\sin(\frac{\pi}{m})$ reikšmės, kai m įgyja įvairias natūraliąsias reikšmes.

Iš vidurinės mokyklos kurso yra gerai žinoma trigonometrinių funkcijų reikšmių lentelė, kai argumento reikšmės yra lygios $\frac{\pi}{3}, \frac{\pi}{4}, \frac{\pi}{6}, \ldots$. Rezultatai yra išreiškiami naudojant baigtinį skaičių sudėties, atimties, daugybos, dalybos ir kvadratinės šaknies traukimo veiksmų su sveikaisiais skaičiais. Aišku, kad tokius kampus nesunku nubrėžti skriestuvu ir liniuote. Šiame darbe parodysime, kaip kompiuterinės algebros sistema Maple gali būti taikoma trigonometrinių funkcijų reikšmėms skaičiuoti.

1 uždavinys. *Į vienetinį apskritimą įbrėžtas taisyklingas septyniolikakampis. Raskite kraštinės ilgį. Rezultatą išreikškite naudodamiesi baigtiniu skaičiumi sudėties, atimties, daugybos, dalybos ir kvadratinės šaknies traukimo veiksmų su sveikaisiais skaičiais.*

Šį uždavinį 1796 metais išsprendė K. Gausas. Jis įrodė (žr. [3], p. 45), kad kraštinė lygi*

$$\frac{1}{4}\sqrt{34-2\sqrt{17}-2\sqrt{34-2\sqrt{17}}-4\sqrt{17+3\sqrt{17}-\sqrt{34-2\sqrt{17}}-2\sqrt{34+2\sqrt{17}}}}.$$

Tais laikais tai buvo sunkus uždavinys ir jo sprendimas buvo rimtas mokslinis atradimas. Apie tai liudija toks faktas, kad Getingene pastatytas Gausui paminklas, kurio pjedestalo skersinis pjūvis yra taisyklingo septyniolikakampio pavidalo. Naudojantis kompiuterinės algebros sistema Maple, šią problemą spręsti visai nesunku.

```
> restart;
> with(geometry,RegularPolygon,point,sides):
```

Įvedame taisyklingą septyniolikakampį

```
> RegularPolygon(DK,17,point(C,0,0),1):
```

ir randame kraštinę a:

```
> a:=sides(DK);
```

$$a := 2\sin\left(\frac{1}{17}\pi\right)$$

Šią formulę lengvai galima gauti be kompiuterio. Toliau randame reikšmę $x = \sin(\frac{\pi}{17})$:

* Knygelėje [3] nurodant šį reiškinį įsivėlė korektūros klaida: vienos šaknies ženklą reikia pratęsti.

```
> x=sin(Pi/17);
```

$$x = \sin\left(\tfrac{1}{17}\pi\right)$$

```
> map(arcsin,%);
```

$$\arcsin(x) = \tfrac{1}{17}\pi;$$

```
> %*17;
```

$$17\arcsin(x) = \pi$$

```
> map(sin,%);
```

$$\sin(17\arcsin(x)) = 0$$

```
> expand(%);
```

$$17x - 816x^3 + 11424x^5 - 71808x^7 + 239360x^9 - 452608x^{11} + 487424x^{13} - 278528x^{15} + 65536x^{17} = 0$$

Gavome algebrinę lygtį

```
> eq:=sort(expand(%/x));
```

$$eq := 65536x^{16} - 278528x^{14} + 487424x^{12} - 452608x^{10} + 239360x^8 - 71808x^6 + 11424x^4 - 816x^2 + 17 = 0$$

Pakeičiame lygtyje x^2

```
> tr := x^2 = 17/32-1/32*sqrt(17)-1/32*sqrt(34-2*sqrt(17))-t;
```

$$tr := x^2 = \tfrac{17}{32} - \tfrac{1}{32}\sqrt{17} - \tfrac{1}{32}\sqrt{34 - 2\sqrt{17}} - t$$

Šis keitimas yra parinktas norit gauti atsakymą, artimą Gauso atsakymui. Toliau pateikiamas bendresnis skaičiavimo būdas, kuriuo gaunamas atsakymas nėra toks kompaktiškas.

```
> tr1:=x=sqrt(rhs(tr));
```

$$tr1 := x = \tfrac{1}{8}\sqrt{34 - 2\sqrt{17} - 2\sqrt{34 - 2\sqrt{17}} - 64t}$$

Toliau atliekame keitimą, skaidome kairiąją pusę į daugiklius ir sprendžiame lygtį. Tarpiniai rezultatatai, kurie yra šiek tiek grėmėzdiški, čia nerodomi. Norint juos pamatyti kompiuterio ekrane, komandų pabaigoje dvitaškį reikia pakeisti kabliataškiu.

```
> simplify(algsubs(tr,eq)):
> factor(%):
> solve(%):
```

Gauname dydžio $x = \sin(\frac{\pi}{17})$ išraišką:

```
> r1:=subs(t=%[1],tr1);
```

$$r1 := x = \tfrac{1}{8}\sqrt{34 - 2\sqrt{17} - 2\sqrt{34 - 2\sqrt{17}} - 2\sqrt{68 - 2\sqrt{17}\sqrt{34 - 2\sqrt{17}} - 6\sqrt{34 - 2\sqrt{17}} + 12\sqrt{17}}}$$

Pažymėjus $q = \sqrt{34 - 2\sqrt{17}}$, šią formulę galima sutrumpinti:

```
> assume(q>0);interface(showassumed=0);
> isolate(q=sqrt(34-2*sqrt(17)),sqrt(17));
```

$$\sqrt{17} = -\tfrac{1}{2}q^2 + 17$$

Todėl

```
> r2:=simplify(subs(%,r1));
```

$$r2 := x = \frac{1}{8}\sqrt{q^2 - 2q - 2\sqrt{272 + q^3 - 40q - 6q^2}}$$

čia $q = \sqrt{34 - 2\sqrt{17}}$.

Naudojantis komanda `evalf`, galima patikrinti ir gauti atsakymą norimu tikslumu. Pavyzdžiui, pasirinkę 20-ies ženklų tikslumą, gauname:

```
> evalf(simplify(subs(q=sqrt(34-2*sqrt(17)),rhs(r2))),20);
```

$$.18374951781657033159$$

```
> evalf(sin(Pi/17),20);
```

$$.18374951781657033158$$

Matome, kad 19 ženklų po kablelio sutampa. Todėl gauta dydžio $x = \sin(\frac{\pi}{17})$ išraiška radikalais yra teisinga.

Gauname, kad septyniolikakampio kraštinė yra lygi

```
> ats:=subs(sin(Pi/17)=rhs(r2),a);
```

$$ats := \frac{1}{4}\sqrt{q^2 - 2q - 2\sqrt{272 + q^3 - 40q - 6q^2}}$$

čia $q = \sqrt{34 - 2\sqrt{17}}$.

Norėdami įsitikinti, ar mūsų atsakymas `ats` sutampa su Gauso atsakymu `G_ats`, patikriname, ar skirtumas `ats-G_ats` lygus nuliui:

```
> G_ats:=sqrt(34-2*sqrt(17)-2*sqrt(34-2*sqrt(17))
  -4*sqrt(17+3*sqrt(17)-sqrt(34-2*sqrt(17))
  -2*sqrt(34+2*sqrt(17)))))/4:
> ATS:=simplify(subs(q=sqrt(34-2*sqrt(17)),ats)):
> normal(ATS-G_ats,expanded):
> radnormal(%,rationalized);
```

$$0$$

Matome, kad skirtumas tikrai lygus nuliui, taigi gavome tą patį, kaip ir Gauso, atsakymą.

$$\textit{Atsakymas. } a = \frac{\sqrt{q^2-2q-2\sqrt{272+q^3-40q-6q^2}}}{4}; \quad \text{čia } q = \sqrt{34 - 2\sqrt{17}}.$$

2 uždavinys. *Apskaičiuokite tas trigonometrinių funkcijų* $\sin(\frac{k\pi}{n})$, $\cos(\frac{k\pi}{n})$, *k ir n — natūralieji skaičiai,* $n \leqslant 64$, *reikšmes, kurias galima išreikšti baigtinį skaičių kartų pasinaudojus sveikųjų skaičių sudėties, atimties, daugybos, dalybos ir kvadratinės šaknies traukimo veiksmais.*

```
> restart;
```

Nubrėžti liniuote ir skriestuvu taisyklingąjį daugiakampį galima, kai kraštinių skaičius (imant iki 64) yra (žr. [3]):

```
> K:=3,4,5,6,8,10,12,15,16,17,20,24,30,32,34,40,48,60,64:
```

Su šiomis vardiklio n reikšmėmis ir pavyksta išspręsti šį uždavinį. Apibendrindami 1 uždavinio sprendimo metodą, parašome procedūrą `valtr`, kuri padeda minėtas reikšmes apskaičiuoti:

```
> valtr:=proc(S)
```

```
option `Copyright Aleksas Domarkas, 2001`;
local f,x,d,spr,ats,k,R;
global K;
if not has(S,Pi) then RETURN(S) end if;
R:=indets(S,function)[1];
f:=op(0,R);
map(invfunc[f],x=R);
%*denom(rhs(%));
if not has({K},denom(op(1,R)))
then RETURN(S) end if;
map(f,%);expand(%);
if has(%,sqrt(1-x^2)) then expand(%/sqrt(1-x^2)) end if;
factor(rhs(%)-lhs(%));
factor(%,{sqrt(17),sqrt(5)});
_EnvExplicit:=true;
spr:=solve({%%,abs(R-x)<10^(-6)},x);
if has(%,RootOf) then R else subs(spr,x) end if;
RETURN(subs(R=%,S));
end proc:
```

Procedūra `valtr` taikoma šitaip:

```
> valtr(cos(Pi/16));
```

$$\frac{1}{2}\sqrt{2+\sqrt{2+\sqrt{2}}}$$

```
> r1:=valtr(sin(Pi/60));
```

$$r1 := \frac{1}{4}\sqrt{8-2\sqrt{7+\sqrt{5}+\sqrt{30+6\sqrt{5}}}}$$

```
> valtr(5*tan(11*Pi/24));
```

$$10+5\sqrt{6}+5\sqrt{3}+5\sqrt{2}$$

Iš dalies šis uždavinys jau yra išspręstas Maple sistemoje. Pavyzdžiui:

```
> r2:=convert(sin(Pi/60),radical);
```

$$r2 := \left(-\frac{1}{16}\sqrt{2}-\frac{1}{8}\sqrt{5+\sqrt{5}}+\frac{1}{16}\sqrt{2}\sqrt{5}\right)\sqrt{3}-\frac{1}{16}\sqrt{2}+\frac{1}{8}\sqrt{5+\sqrt{5}}+\frac{1}{16}\sqrt{2}\sqrt{5}$$

```
> convert(5*tan(11*Pi/24),radical);
```

$$5\sqrt{2}\sqrt{3}+5\sqrt{3}+5\sqrt{2}+10$$

Bet ši konvertavimo procedūra neveikia, kai argumento reikšmės lygios $\frac{\pi}{16}$, $\frac{\pi}{17}$, $\frac{\pi}{32}$, $\frac{\pi}{34}$, $\frac{\pi}{48}$, $\frac{\pi}{64}$. Atsakymo išraiška gali būti gana skirtinga. Patikrinti atsakymus galima komanda `evalf`. Pavyzdžiui:

```
evalf(sin(Pi/60),30); evalf(r1,30); evalf(r2,30);
```

$$.523359562429438327221186296092$$
$$.523359562429438327221186295998$$
$$.52335956242943832722118629609$$

Konvertavimo procedūros tekstą galima pamatyti įvedus

```
> interface(verboseproc=2);
> eval('convert/radical'):
```

Šio teksto čia nerodome (gale dvitaškis), nes jis pakankamai ilgas. Be to, konvertavimo procedūra remiasi dar kitomis paprogramėmis.

Sudarome visų koeficientų $\frac{k}{n}$ seką:

```
> K_SEKA:=NULL:
> for k in [K] do
  for i to k-1 do
  if not has([K_SEKA],i/k)
  then K_SEKA:=K_SEKA,i/k; end if;
  end do; end do;
```

Šios sekos narių skaičius yra

```
> m:=nops([K_SEKA]);
```

$$m := 191$$

```
> K_SEKA[1..10],' ... ',K_SEKA[m-2..m];
```

$$\frac{1}{3}, \frac{2}{3}, \frac{1}{4}, \frac{1}{2}, \frac{3}{4}, \frac{1}{5}, \frac{2}{5}, \frac{3}{5}, \frac{4}{5}, \frac{1}{6}, \ldots, \frac{59}{64}, \frac{61}{64}, \frac{63}{64}$$

Taupydami vietą ir laiką, atsakymus pateikiame tik sin funkcijų, kurių argumento koeficientai yra iš dalinės sekos

```
> DS:=1/5,1/8,1/10,1/12, 1/15, 1/16, 1/20, 1/24, 1/30, 1/32,
1/40, 1/48, 1/60, 1/64:
> for k in [DS] do sin(k*pi)=valtr(sin(k*Pi)); end do;
```

$$\sin\left(\tfrac{1}{5}\pi\right) = \tfrac{1}{4}\sqrt{10-2\sqrt{5}} \qquad \sin\left(\tfrac{1}{8}\pi\right) = \tfrac{1}{2}\sqrt{2-\sqrt{2}}$$

$$\sin\left(\tfrac{1}{12}\pi\right) = \tfrac{1}{4}\sqrt{6} - \tfrac{1}{4}\sqrt{2} \qquad \sin\left(\tfrac{1}{10}\pi\right) = \tfrac{1}{4}\sqrt{5} - \tfrac{1}{4}$$

$$\sin\left(\tfrac{1}{16}\pi\right) = \tfrac{1}{2}\sqrt{2-\sqrt{2-\sqrt{2+\sqrt{2}}}} \qquad \sin\left(\tfrac{1}{15}\pi\right) = \tfrac{1}{4}\sqrt{7-\sqrt{5}-\sqrt{30-6\sqrt{5}}}$$

$$\sin\left(\tfrac{1}{24}\pi\right) = \tfrac{1}{4}\sqrt{8-2\sqrt{6-2\sqrt{2}}} \qquad \sin\left(\tfrac{1}{20}\pi\right) = \tfrac{1}{4}\sqrt{8-2\sqrt{10+2\sqrt{5}}}$$

$$\sin\left(\tfrac{1}{32}\pi\right) = \tfrac{1}{2}\sqrt{2-\sqrt{2+\sqrt{2+\sqrt{2}}}} \qquad \sin\left(\tfrac{1}{30}\pi\right) = -\tfrac{1}{8} - \tfrac{1}{8}\sqrt{5} + \tfrac{1}{8}\sqrt{30-6\sqrt{5}}$$

$$\sin\left(\tfrac{1}{48}\pi\right) = \tfrac{1}{4}\sqrt{8-2\sqrt{8+2\sqrt{6}+2\sqrt{2}}} \qquad \sin\left(\tfrac{1}{40}\pi\right) = \tfrac{1}{4}\sqrt{8-2\sqrt{8+2\sqrt{10+2\sqrt{5}}}}$$

$$\sin\left(\tfrac{1}{64}\pi\right) = \tfrac{1}{2}\sqrt{2-2\sqrt{2+\sqrt{2+\sqrt{2+\sqrt{2}}}}} \qquad \sin\left(\tfrac{1}{60}\pi\right) = \tfrac{1}{4}\sqrt{8-2\sqrt{7+\sqrt{5}+\sqrt{30+6\sqrt{5}}}}$$

Čia nepateikėme kampų $\frac{\pi}{17}$ ir $\frac{\pi}{34}$ reikšmių, nes formulės yra ilgos. Šį bei kitus darbus galima rasti internete [2]. Atidaryti ir skaičiavimus atlikti galima su Maple 6 arba Maple 7 programų versijomis. Manau, kad pateiktos programos nėra visiškai tobulos. Siūlau skaitytojui jas patobulinti ir išbandyti.

1. A. Domarkas, Maple — naujos galimybės, *Alfa plius omega*, **3**, 88–97, 2001.
2. A. Domarkas, *Matematikos praktikumas su Maple*, http://ieva.maf.vu.lt/home/aleksas
3. J. Kubilius, *Antanas Baranauskas ir matematika*, MII, Vilnius, 2001.